
Abstract. Refactoring is a popular technique from the OO-community to re-
structure code: it aims at improving software readability, maintainability and ex-
tensibility. In this paper we apply the ideas of refactoring to Prolog programs. We
present a catalogue of refactorings adapted to or specificly developed for Prolog.
We also discuss ViPReSS, our semi-automatic refactoring browser, and our ex-
perience with applying ViPReSS to a large Prolog legacy system. Our conclusion
is that refactoring is clearly both viable and desirable in the context of Prolog.

1 Introduction

Logic programming languages and refactoring have already been put together
at different levels. Tarau [18] calls his re-implementation of the Prolog language
refactoring. However, this approach differs significantly from the traditional no-
tion of refactoring as introduced by Fowler [4]. We follow the latter definition.

Recent relevant work is [19] in the context of object oriented languages: a
meta logic very similar to Prolog is used to detect for instance obsolete para-
meters. Refactoring of such code consists in removing obsolete parameters and
also the problem of cascading refactoring is discussed.

None of these papers, however, considers applying refactoring techniques
to logic programs. In our previous paper [16] we have emphasised the import-
ance of refactoring for logic programming and discussed the applicability of
the refactoring techniques developed for object-oriented languages [3] to Pro-
log and CLP-programs. Seipel et al. [15] include refactoring among the analysis
and visualisation techniques that can be easily implemented by means of FN-
QUERY, a Prolog-inspired query language for XML. However, the discussion
stays at the level of an example and no detailed study has been conducted.

The outline of this paper is as follows. We start with an overview of the gen-
eral principles of refactoring in Section 2. Next we illustrate the use of several

refactoring techniques on a small example in Section 3. Then a more compre-
hensive catalogue of refactorings for Prolog programs is given in Section 4. In
Section 5 we introduce ViPReSS , our refactoring browser, currently implement-
ing most of the refactorings of the catalogue. ViPReSS has been successfully
applied for refactoring a 50,000 lines-long legacy system developed and used at
K.U.Leuven to manage educational activities of the Computer Science depart-
ment. Finally, in Section 6 we conclude.

2 Overview of Refactoring

Program changes take up a substantial part of the entire programming effort.
Often changes are required to incorporate additional functionality or to improve
efficiency. In both cases, a preliminary step of improving the design without
altering the external behaviour is recommended. This methodology, called re-
factoring, emerged from a number of pioneer results in the OO-community [4,
11, 13] and recently came to prominence for functional languages [9]. More
formally, refactoring is a source-to-source program transformation that changes
program structure and organisation, but not program functionality. The major
aim of refactoring is to improve readability, maintainability and extensibility of
the existing software. While performance improvement is not considered as a
crucial issue for refactoring, it should be noted that well-structured software is
more amenable to performance tuning. We also observe that certain techniques
that were developed in the context of program optimisation can improve pro-
gram organisation and, hence, can be considered refactoring techniques.

To achieve the goals above two questions have to be answered, namely,
where and how transformations need to be performed. Unlike automated pro-
gram transformations, neither of the steps aims at transforming the program
fully automatically. The decision whether the transformation should be applied
is left to the program developer. However, providing automated support for re-
factoring is a useful and important challenge.

Deciding automatically where to apply a transformation can be a difficult
task on its own. A number of ways to resolve this may be considered. First, pro-
gramming analysis approaches can be used. For example, it is common prac-
tice while ordering predicate arguments to start with the input arguments and
end with the output arguments. Mode information can be used to detect when
this rule is violated and to suggest the user to reorder the arguments. Second,
one can try and predict further refactorings based on the program transforma-
tions already applied. For instance, if constraints have been simplified by us-
ing De Morgan’s rules, the system can propose to apply the same rules to re-
write additional constraints. Eventually useful sequences of refactoring steps

can be learned analogously to the automated macro construction [7]. By fol-
lowing these approaches the automatic refactoring tools, so called refactoring
browsers, can be expected to make suggestions on when refactoring transform-
ations should be applied. These suggestions should then be either confirmed or
rejected by the program developer.

Answering how the program should be transformed might also require the
user’s input. We illustrate this point by observing that one possible refactoring
renames a predicate: while automatic tools can hardly be intelligent enough to
guess the new predicate name, they should be able to detect all program points
affected by the change. Moreover, assuming certain properties, such as absence
of the user-defined meta-predicates, on the original program can lead to a more
readable and compact transformed program. In this case the user is requested to
confirm that the properties in question indeed hold.

3 Detailed Prolog Refactoring Example

We illustrate some of the techniques proposed by a detailed refactoring example.
Consider the following code fragment borrowed from O’Keefe’s “The Craft of
Prolog” [10], p. 195. It describes three operations on a reader data structure used
to sequentially read terms from a file. The three operations are make reader/3
to initialise the data structure, reader done/1 to check whether no more terms
can be read and reader next/3 to get the next term and advance the reader.

O’Keefe’s orignal version
make_reader(File,Stream,State) :-

open(File,read,Stream),
read(Stream,Term),
reader_code(Term,Stream,State).

reader_code(end_of_file,_,end_of_file) :- ! .
reader_code(Term,Stream,read(Term,Stream,Position)) :-

stream_position(Stream,Position).

reader_done(end_of_file).

reader_next(Term,read(Term,Stream,Pos),State)) :-
stream_position(Stream,_,Pos),
read(Stream,Next),
reader_code(Next,Stream,State).

We will now apply several refactorings to the above program to improve its
readability.

First of all, we use if-then-else introduction to get rid of the ugly red cut in
the reader code/3 predicate:

Replace cut by if-then-else
reader_code(Term,Stream,State) :-

( Term = end_of_file,
State = end_of_file ->

true
;

State = read(Term,Stream,Position),
stream_position(Stream,Position)

).

This automatic transformation reveals two malpractices, the first of which
is producing output before the commit, something O’Keefe himself disapproves
of (p. 97). This is fixed manually to:

Output after commit
reader_code(Term,Stream,State) :-

( Term = end_of_file ->
State = end_of_file

;
State = read(Term,Stream,Position),
stream_position(Stream,Position)

).

The second malpractice is a unification in the condition of the if-then-else
where actually an equality test is meant. Consider that the Term argument is
a variable. Then the binding is certainly unwanted behaviour. Manual change
generates the following code:

Equality test
reader_code(Term,Stream,State) :-

( Term == end_of_file ->
State = end_of_file

;
State = read(Term,Stream,Position),
stream_position(Stream,Position)

).

Next, we notice that the sequence read/2, reader code/3 occurs twice,
either by simple observation or by computing common body subsequences. By
applying predicate extraction of this common sequence, we get:

Predicate extraction
make_reader(File,Stream,State) :-

open(File,read,Stream),
read_next_state(Stream,State).

reader_next(Term,read(Term,Stream,Pos),State)) :-
stream_position(Stream,_,Pos),
read_next_state(Stream,State).

read_next_state(Stream,State) :-

read(Stream,Term),
reader_code(Term,Stream,State).

Next we apply O’Keefe’s own principle of putting the input argument first
and the output arguments last (p. 14–15):

Argument reodering
reader_next(read(Term,Stream,Pos),Term,State) :-

stream_position(Stream,_,Pos),
read_next_code(Stream,State).

To finish, we introduce less confusing and overlapping names for the read/3
functor, the stream position/[2,3] built-ins and a more consistent naming
for make reader, more in line with the other two predicates in the interface.
O’Keefe stresses the importance of consistent naming conventions (p. 213).

Note that direct renaming of built-ins such as stream position is not pos-
sible, but with some cleverness a similar effect can be achieved: simply extract
the built-in into a new predicate with the desired name.

Renaming
reader_init(File,Stream,State) :-

open(File,read,Stream),
reader_next_state(Stream,State).

reader_next(reader(Term,Stream,Pos),Term,State)) :-
set_stream_position(Stream,Pos),
reader_next_state(Stream,State).

reader_done(end_of_file).

reader_next_state(Stream,State) :-
read(Stream,Term),
build_reader_state(Term,Stream,State).

build_reader_state(Term,Stream,State) :-
( Term == end_of_file ->

State = end_of_file
;

State = reader(Term,Stream,Position),
get_stream_position(Stream,Position)

).

set_stream_position(Stream,Position) :-
stream_position(Stream,_,Position).

get_stream_position(Stream,Position) :-
stream_position(Stream,Position).

While the above changes could be perfomed manually, a refactoring browser
such as ViPReSS (see Section 5) guarantees consistency, correctness and further-
more can automatically single out opportunities for refactoring.

4 Comprehensive Catalogue of Prolog refactorings

In this section we present a number of refactorings that we have found to be
useful when Prolog programs are considered. A more comprehensive discussion
of the presented refactorings can be found in [14].

We stress that the programs are not limited to pure logic programs, but may
contain various built-ins such as those defined in the ISO standard [1]. The only
exception are higher-order constructs that are not dealt with automatically, but
manually. Automating the detection and handling of higher-order predicates is
an important part of future work.

The refactorings in this catalogue are grouped by scope. The scope expresses
the user-selected target of a particular refactoring. While the particular refactor-
ing may affect code outside of the selected scope, it is only because the refact-
oring operation detects a dependency outside the scope.

For Prolog programs we distinguish the following four scopes, based on the
code units of Prolog. It should be noted that a similar distinction can be made
for other languages based on their code units.

The system scope encompasses the entire code base. Hence the user does not
want to transform a particular subpart, but to affect the system as a whole. The
system scope is obviously appropriate for all programming paradigms. It even
applies to multi-language systems.

System Scope Refactorings
• Extract common code into predicates
• Hide predicates (remove them from export lists)
• Remove dead code
• Remove duplicate predicates
• Remove redundant arguments
• Rename functor

The module scope considers a particular module. Usually a module is imple-
menting a well-defined functionality and is typically contained in one file. This
scope corresponds to refactorings operating on classes in Java or modules in
Haskell. Module Scope Refactorings

• Merge modules
• Remove dead code intra-module
• Rename module
• Split module

The predicate scope targets a single predicate. For example, the moving of a
predicate to a different module, affects the predicate directly. The code that de-
pends on the predicate may need updating as well. But this is considered an

implication of the refactoring of which either the user is alerted or the neces-
sary transformations are performed implicitly. The counterparts of predicates in
Prolog are methods in Java and functions in Haskell.

Predicate Scope Refactorings
• Add argument
• Move predicate
• Rename predicate
• Reorder arguments

The clause scope affects a single clause in a predicate. Usually, this does not
affect any code outside of the clause directly. An example is when a part of
the clause body is replaced by a call to a new predicate that consists of the
original goal. Code outside of the clause scope is affected in as much as that the
new predicate is introduced and the necessary import declarations are added or
removed. In object-oriented and functional programming languages this scope
affects methods and function bodies respectively.

Clause Scope Refactorings
• Extract predicate locally
• Invert if-then-else (negate condition and reorder branches)
• Replace cut by if-then-else
• Replace unification by (in)equality test
• Unfold goal

5 The ViPReSS refactoring browser

The refactoring techniques presented above have been implemented in the ViPReSS
refactoring browser. To facilitate acceptance of the tool ViPReSS by the de-
velopers community has been implemented not as a stand-alone application but
on the basis of VIM (http://www.vim.org/), a popular clone of the well-
known VI editor. Techniques like predicate duplication provided by refactoring
browsers such as HaRe [9] are already included in the editor facilities of VIM.

Most of the refactoring tasks have been implemented as SICStus Prolog
programs inspecting source files and/or call graphs. Updates to files have been
implemented either directly in the scripting language of VIM or, in the case
many files had to be updated at once, through ed scripts. VIM functions to
initiate the refactorings and to get user input have been written.

ViPReSS has been successfully applied to a large (more than 53,000 lines)
legacy system used at the Computer Science department of the Katholieke Uni-
versiteit Leuven to manage the educational activities. The system, called BTW
(Flemish for value-added tax), has been developed and extended since the early

eighties by more than ten different programmers, many of whom are no longer
employed by the department. The implementation has been done in the Master-
ProLog [6] system that will be no longer supported from spring 2004.

By using the refactoring techniques we aimed to obtain a better understand-
ing of this real-world system, to improve its structure and maintainability, and
to prepare it for further intended changes such as porting it to a state-of-the-art
Prolog system and adapting it to new educational tasks the department is facing
as a part of the unified Bachelor-Master system in Europe.

We started by removing some parts of the system that have been identified
by the expert as obsolete, including out-of-fashion user interfaces and outdated
versions of program files. The bulk of dead code was eliminated in this way,
reducing the system size to a mere 20,000 lines. Note that dead code elimination
could have been applied after simply removing the outdated toplevel predicates.

Next, we applied most of the system-scope refactorings described above.
Even after removal of dead code by the experts ViPReSS identified and elim-
inated 299 dead predicates. This reduced the the size by another 1,500 lines.
Moreover ViPReSS discovered 79 pairwise identical predicates. In most of the
cases, identical predicates were moved to new modules used by the original
ones. The previous steps allowed us to improve the overall structure of the pro-
gram by reducing the number of system files from 294 to 116. The size of the
system has simultaneously diminished to 18,000 lines. Very little time was spent
to bring the system into this state. The experts were sufficiently familiar with the
system to immediately identify obsolete parts. The system-scope refactorings
took only a few minutes each.

The second step of refactoring consisted of a thorough code inspection
aimed at local improvement. Many malpractices have been identified: excess-
ive use of cut combined with producing the output before commit being the
most notorious one. Additional “bad smells” discovered include bad predicate
names such as q, unused arguments and unifications used instead of identity
checks or numerical equalities. Some of these were located by ViPReSS , others
were recognised by the users, while ViPReSS was used to perform the corres-
ponding transformations. This step is more demanding of the user. She has to
consider all potential candidates for refactoring separately and decide on what
transformations apply. Hence, the lion’s share of the refactoring time is spent on
these local changes.

In summary, from the case study we learned that automatic support for re-
factoring techniques is essential and that ViPReSS is well-suited for this task. As
the result of applying refactoring to BTW we obtained better-structured lumber-
free code. It is now not only more readable and understandable but it also sim-
plifies implementing the intended changes. Considering the relative time invest-

ments of the global and the local refactorings, we recommend to start out with
the global ones and then selectively apply local refactorings as the need occurs.

6 Conclusions and Future Work

In this paper we have shown that the ideas of refactoring, originating in the OO-
community, are applicable and important for logic programming. Refactoring
helps bridging the gap between prototypes and real-world applications. Indeed,
extending a prototype to provide additional functionality often leads to cumber-
some code. Refactoring allows software developers both to clean up code after
changes and to prepare code for future changes.

We have presented a catalogue of refactorings at different scopes of a pro-
gram, containing both previously known refactorings for object-oriented lan-
guages now adapted for Prolog and entirely new Prolog-specific refactorings.
Although the presented refactorings do require human input to make certain
choices, as it is in the general spirit of refactoring, a large part of the work
can be automated. Our refactoring browser ViPReSS integrates the automatable
parts of the presented refactorings in the VIM editor and can be downloaded
from http://www.cs.kuleuven.ac.be/˜toms/vipress.

In the logic programming community questions related to refactoring have
been intensively studied in context of program transformation and specialisa-
tion [2, 8, 12]. There are two important differences with this line of work. First
of all, refactoring does not aim at optimising program performance but at im-
proving program readability, maintainability and extensibility. Second, user in-
put is essential in the refactoring process while traditionally only automatic ap-
proaches were considered. Moreover, usually program transformations are part
of a compiler and hence, they are “invisible” to the program developer. How-
ever, it should be noted that some of the transformations developed for program
optimisation can be considered as refactorings and should be implemented in
refactoring browsers. This is the case, for instance, for dead code elimination
and removing redundant arguments.

To further increase the level of automation of particular refactorings addi-
tional information such as types and modes can be used. This allows for ex-
ample to reorder the arguments in a predicate putting the output variables last
as recommended by [10], or to improve the output of if-then-else inversion. To
obtain this information the refactoring system could be extended with type and
mode analyses. On the other hand, it seems worthwhile to consider the proposed
refactorings in the context of languages with type and mode declarations as Mer-
cury [17] and HAL [5], especially as these languages claim to be of greater in-
dustrial relevance than traditional Prolog. Moreover, dealing with higher order

is essential for refactoring in a real world context. The above mentioned lan-
guages with explicit declarations for such constructs would again facilitate the
implementation of an industrial strength refactoring environment.

